* Appendix;
/*This appendix contains the macro ORDINAL, that produces:
I) cut-point specific betas and standard errors for odds ratios contributing to the proportional odds ratio (crude and adjusted);
ii) betas and standard errors for proportional odds ratios (crude and adjusted) and p-value for test of homogeneity;
iii) cut-point specific betas and standard errors for odds ratios contributing to the continuation ratio (adjusted only); and
iv) beta and standard error for continuation ratio (adjusted) and log likelihoods necessary to perform test of homogeneity.
The following parameters are required to run this macro:
i) "SASdata" is the name of the input SAS data set;
ii) "Outcome" is the name of the outcome variable. It should be coded 0=least severe category to k=most severe category;
iii) "Tcutpts" is the total number of cut-points, equivalent to 1 less than the total number of levels of the outcome variable;
iv) "Exposure" is the name of the exposure variable. It should be coded
1=exposed or 0=unexposed, or can be continuous. The macro can be adjusted to accept categorical levels; and
v) "Confound" is a list of confounding variables.
These parameters should be entered in the following macro call, and placed at the end of the program:
%ORDINAL(SASdata,outcome,tcutpts,exposure,confound);
*/
%MACRO ORDINAL(SASdata,outcome,tcutpts,exposure,confound);
* Proportional Odds Model;
%do i=1 %to &tcutpts;
data t;
set &SASdata;
if .<&outcome<&i then dvar=0;
else if &outcome>. then dvar=1;
label dvar="Dichotomized outcome based on cutpoint";
run;
proc logistic descending;
model dvar=&exposure/risklimits;
title "&outcome: Crude OR &i contributing to Cumulative OR";
run;
proc logistic descending;
model dvar=&exposure &confound/risklimits;
title "&outcome: Adjusted OR &i contributing to Cumulative OR";
run;
%end;
proc datasets;
delete t;
run;
proc logistic data=&SASdata descending;
model &outcome=&exposure/risklimits;
title "&outcome: Crude Cumulative OR";
run;
proc logistic data=&SASdata descending;
model &outcome=&exposure &confound/risklimits;
title "&outcome: Adjusted Cumulative OR";
run;
* Continuation Ratio Model;
data restruct;
set &SASdata;
do i=1 to &tcutpts;
if &outcome>=(i-1) then do;
ORi=i;
array ctpt(*) ctpt1-ctpt&tcutpts;
array ctptint (*) ctptint1-ctptint&tcutpts;
ctpt(i)=1;
do j=1 to &tcutpts;
if ctpt(j)=. then ctpt(j)=0;
ctptint(j)=&exposure*ctpt(j);
end;
if &outcome=(i-1) then dvar=0;
else if &outcome>. then dvar=1;
label dvar="Dichotomized outcome based on cutpoint";
label ctpt1="Indicator of Cutpoint 1";
label ctpt&tcutpts="Indicator of Cutpoint &tcutpts";
output;
end;
end;
run;
proc sort data=restruct;
by ORi;
run;
proc logistic data=restruct descending;
by ORi;
model dvar=&exposure &confound/risklimits;
title "&outcome: Adjusted ORi contributing to Continuation OR";
run;
proc logistic data=restruct descending;
model dvar=&exposure ctpt2-ctpt&tcutpts &confound/risklimits;
title "&outcome: Adjusted Continuation OR";
run;
proc logistic data=restruct descending;
model dvar=&exposure ctpt2-ctpt&tcutpts ctptint2-
ctptint&tcutpts &confound;
title "&outcome: 2nd log likelihood needed to test
homogeneity";
run;
%MEND;
/*
Example: To obtain: (i) proportional and cut-point specific binary odds ratios, with 95% confidence limits, and global test of homogeneity, and (ii) continuation and cut-point specific binary odds ratios, with 95% confidence limits, and log likelihoods necessary to test homogeneity from: (i) a SAS data set called tt, (ii & iii) a 6-leveled ordinal outcome variable called pain (coded 0=no pain to 5=continuous pain), (iv) an exposure variable called status, and (iv) controlling for sex (coded 0-1), age (continuous) and smoking status (2 dummy variables, smk1 & smk2), submit the following call statement:
%ORDINAL(tt,pain,5,status,%STR(sex age smk1-smk2));
*/
======================================================================
Ordinal regression models can also be fit using SPlus, via a set of  functions written by Frank Harrel.
The following is excerpted from correspondance on the Splus list server regarding this.
 From fharrell@virginia.eduMon Oct 5 10:35:13 1998
Date: Thu, 1 Oct 1998 09:20:13 -0400
Subject: Re: [S] Proportional odds model
The proportional odds model is due to Walker & Duncan (1967 Biometrika) although there are other relevant papers dating back to the 1930s. You can fit this model with the lrm function in the Design library (see web site below or statlib). The discrete proportional hazards model is the continuation ratio ordinal logistic model, which can also be fit using lrm. See my article in Statistics in Medicine, April 1998. Note that Design requires the Hmisc library also. 
======================================================================
 

